Knowledge Representation and Semantic Technologies

Exercises on Datalog and ASP

Riccardo Rosati

Corso di Laurea Magistrale in Ingegneria Informatica
Sapienza Università di Roma
2015/2016

Exercise 1

Given the following positive Datalog program P:

```
r(x,y) :- s(x,y).
r(x,y) :- r(x,z), s(z,y).
t(x) :- r(x,x).
q(y) :- t(x), r(x,y).
s(a,b).
s(b,c).
s(c,a).
```

1) compute the minimal model of P;
2) tell if atom q(a) is entailed by P.

Exercise 1 - Solution

To compute the minimal model, we use the semi-naive evaluation method. We first define the program P' with Δ-relations:

Δr(x,y) :- s(x,y). [rule R1]
Δr(x,y) :- Δr(x,z), s(z,y). [rule R2]
Δt(x) :- Δr(x,x). [rule R3]
Δq(y) :- Δt(x), r(x,y). [rule R4]
Δq(y) :- t(x), Δr(x,y). [rule R5]
s(a,b). s(b,c). s(c,a).

Exercise 1 - Solution

We then execute the iterative computation of the relations r, t, q through semi-naive evaluation on P':

Step 1:

Δr = { (a,b), (b,c), (c,a) } (using rule R1),
Δt = {}, Δq = {}
r = { (a,b), (b,c), (c,a) }, t={}, q={}

Step 2:

Δr = { (a,c), (b,a), (c,b) } (using rule R2),
Δt = {}, Δq = {}
r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b) }, t={}, q={}

Exercise 1 - Solution

Step 3:

Δr = { (a,a), (b,b), (c,c) } (using rule R2),

Δt = {}, Δq = {}

r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b), (a,a), (b,b), (c,c) },

t={}, q={}

Step 4:

Δr = {}, Δt = {a,b,c} (using rule R3),

Δq = {}

r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b), (a,a), (b,b), (c,c) },

t={a,b,c}, q={}

Exercise 1 - Solution

Step 5:

$\Delta r = \{\}$, $\Delta t = \{\}$, $\Delta q = \{a,b,c\}$ (using rule R4)

r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b), (a,a), (b,b), (c,c) },

t={a,b,c}, q={a,b,c}

Step 6:

$\Delta r = \{\}$, $\Delta t = \{\}$, $\Delta q = \{\}$

Exercise 1 - Solution

The minimal model of P is thus the following:

s = { (a,b), (b,c), (c,a) }

r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b), (a,a), (b,b), (c,c)}

t = {a,b,c}

q = {a,b,c}

Exercise 1 - Solution

Finally, since atom q(a) belongs to the minimal model of P, it is entailed by P.

Exercise 2

Given the following positive Datalog program with constraints P':

```
r(x,y) :- s(x,y).
r(x,y) :- r(x,z), s(z,y).
t(x) :- r(x,x).
q(y) :- t(x), r(x,y).
:- t(x), q(x).
s(a,b). s(b,c). s(c,a).
```

compute the minimal model of P'.

Exercise 2 - Solution

We notice that the program P' is the same as the positive program of Exercise 1, plus the constraint :- t(x), q(x). Namely, $P' = P \cup \{ \text{:-}t(x), q(x) \}$

So, to answer the question we only have to check whether the minimal model of P satisfies such a constraint.

Exercise 2 - Solution

The minimal model M of P (see Exercise 1) is:

s = { (a,b), (b,c), (c,a) }
r = { (a,b), (b,c), (c,a), (a,c), (b,a), (c,b), (a,a), (b,b), (c,c)}
t = {a,b,c}
q = {a,b,c}

M does not satisfy the constraint :- t(x), q(x).
(e.g., both t(a) and q(a) belong to M).

So, we conclude that there exists no (minimal) model for P.

Exercise 3

Given the following ASP program P:

```
r(x,y) :- s(x,y).
r(x,y) :- r(x,z), s(z,y).
t(x,y) :- r(x,y), not s(x,y).
s(a,b).  s(b,c).
```

1) tell whether P is stratified;
2) compute the answer sets of P.

Exercise 3 - Solution

We start by observing that there are no negated atoms involving IDB predicates (the only negated atom is relative to the EDB predicate s).

Therefore, the labeled dependency graph of P does not contain any negated edge, and hence no cycle containing a negated edge.

Consequently, program P is stratified.

Exercise 3 - Solution

Since P is stratified, it has only one answer set, and we can compute such an answer set through semi-naive evaluation of each single stratum of the program (P has actually only one stratum).

In the computation, we have to evaluate the negated atom not s(x,y) as a positive atom over the new EDB predicate not-s whose extension is the set of tuples of constants that do not belong to s:

not-s = { (a,a), (a,c), (b,a), (b,b), (c,a), (c,b), (c,c) }

Exercise 3 - Solution

The rules with Δ-relations are the following:

$\Delta r(x,y)$:- $s(x,y)$.

$\Delta r(x,y)$:- $\Delta r(x,z)$, $s(z,y)$.

$\Delta t(x,y)$:- $\Delta r(x,y)$, not-$s(x,y)$.

Step 1: $\Delta r = \{ (a,b), (b,c) \}$, $\Delta t = \{\}$

Step 2: $\Delta r = \{ (a,c) \}$, $\Delta t = \{\}$

Step 3: $\Delta r = \{\}$, $\Delta t = \{ (a,c) \}$

Step 4: $\Delta r = \{\}$, $\Delta t = \{\}$

Exercise 3 - Solution

The answer set of P is thus the following:

r = { (a,b), (b,c), (a,c) }

t = { (a,c) }